

Las clases File y FileInfo

Al igual que ocurre con las clases para manejar los directorios, tenemos dos clases diferentes para manejar los ficheros. Una de ellas (*File*) todos los métodos que tiene son estáticos, por tanto podemos usarlos directamente sin crear una nueva instancia de la clase. Por otro lado la clase *FileInfo* es una clase de la que tenemos que crear un nuevo objeto para poder usarla, al crear ese objeto (o instancia), tenemos que indicar el fichero al que hará referencia, ese fichero no tiene porqué existir previamente.

En el siguiente ejemplo puedes ver cómo usar estas dos clases:

```
Imports System
Imports System.IO

Module Module1
    Const fic1 As String = "E:\Pruebas\Prueba.txt"
    Const fic2 As String = "E:\Pruebas\SubDir\Prueba3.txt"

    Sub Main()
        Console.WriteLine("Ejemplo de la clase File:")
        claseFile()
        Console.WriteLine()
        Console.WriteLine("Ejemplo de la clase FileInfo:")
        claseFileInfo()
        Console.WriteLine()
        Console.ReadLine()
    End Sub

    Sub claseFile()
        ' comprobar si el fichero existe
        If File.Exists(fic1) Then
            Console.WriteLine("El fichero '{0}' SI existe", fic1)
            Console.WriteLine("Fecha creación: {0}", File.GetCreationTime(fic1))
        Else
            Console.WriteLine("El fichero '{0}' NO existe", fic1)
        End If
    End Sub

    Sub claseFileInfo()
        Dim fi As New FileInfo(fic2)
        ' Comprobar si existe el fichero:
        If fi.Exists Then
            Console.WriteLine("El fichero '{0}' SI existe", fic2)
            Console.WriteLine("Fecha creación: {0}", fi.CreationTime)
            agregarTexto(fi)
        Else
            Console.WriteLine("El fichero '{0}' NO existe", fic2)
            ' lo creamos
            fi.Create()
        End If
    End Sub

    Sub agregarTexto(ByVal fi As FileInfo)
        Dim fs As StreamWriter
        fs = fi.CreateText
        fs.WriteLine("Hola, Mundo")
        fs.Flush()
        fs.Close()
        fi.Refresh()
        Console.WriteLine("Tamaño : {0}", fi.Length)
    End Sub
End Module
```

Con estas clases, además de obtener información de los archivos sobre los que actúan, podemos realizar tareas comunes, como saber si existe un archivo, copiarlo, moverlo o eliminarlo. También podemos usar métodos que nos permiten abrirlos para leer o escribir en ellos, con métodos que devuelven tanto objetos del tipo `FileStream` como del tipo `StreamWriter`.

Algunos de los métodos de la clase `File` son:

***Copy::** Copia un fichero existente de un nuevo fichero.

***Delete:** Borra un fichero

***Exists:** Determina si un fichero existe

***GetAttributes :** Obtiene los atributos de un fichero

***Move:** Mueve un fichero a una nueva localización

***SetAttributes:** Establece los atributos de un fichero.

.....

Consultar [File](#) y [FileInfo](#)